
21 Offline Belief State Planning

In the worst case, an exact solution for a general finite-horizon POMDP is PSPACE-
complete, which is a complexity class that includes NP-complete problems and
is suspected to include problems that are even more difficult.1 General infinite-

1 C. Papadimitriou and J. Tsitsik-
lis, “The Complexity ofMarkovDe-
cision Processes,” Mathematics of
Operation Research, vol. 12, no. 3,
pp. 441–450, 1987.horizon POMDPs have been shown to be uncomputable.2 Hence, there has been a
2 O. Madani, S. Hanks, and A.
Condon, “On the Undecidability
of Probabilistic Planning and Re-
lated Stochastic Optimization Prob-
lems,”Artificial Intelligence, vol. 147,
no. 1–2, pp. 5–34, 2003.

tremendous amount of research recently on approximationmethods. This chapter
discusses various offline POMDP solution methods, which involve performing
all or most of the computation prior to execution. We focus on methods that
represent the value function as alpha vectors and different forms of interpolation.

21.1 Fully Observable Value Approximation

One of the simplest offline approximation techniques is QMDP, which derives its
name from the action value function associated with a fully observed MDP.3 This

3 M.L. Littman, A.R. Cassandra,
and L. P. Kaelbling, “Learning Poli-
cies for Partially Observable Envi-
ronments: Scaling Up,” in Interna-
tional Conference on Machine Learn-
ing (ICML), 1995. A proof that
QMDP provides an upper bound
on the optimal value function is
given by M. Hauskrecht, “Value-
Function Approximations for Par-
tially Observable Markov Decision
Processes,” Journal of Artificial Intel-
ligence Research, vol. 13, pp. 33–94,
2000.

approach, as well as several others discussed in this chapter, involve iteratively
updating a set Γ of alpha vectors, as shown in algorithm 21.1. The resulting set Γ

defines a value function and a policy that can be used directly or with one-step
lookahead as discussed in the previous chapter, though the resulting policy will
only be an approximation of the optimal solution.

function alphavector_iteration(𝒫::POMDP, M, Γ)
for k in 1:M.k_max

Γ = update(𝒫, M, Γ)
end
return Γ

end

Algorithm 21.1. Iteration structure
for updating a set of alpha vec-
tors Γ used by several of the meth-
ods in this chapter. The various
methods, including QMDP, differ
in their implementation of update.
After k_max iterations, this function
returns a policy represented by the
alpha vectors in Γ.

QMDP (algorithm 21.2) constructs a single alpha vector αa for each action a

using value iteration. Each alpha vector is initialized to zero, and then we iterate:

α
(k+1)
a (s) = R(s, a) + γ ∑

s′
T(s′ | s, a)max

a′
α
(k)
a′ (s

′) (21.1)

Each iteration requires O(|A|2|S|2) operations. Figure 21.1 illustrates the process.

struct QMDP
k_max # maximum number of iterations

end

function update(𝒫::POMDP, M::QMDP, Γ)
𝒮, 𝒜, R, T, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.R, 𝒫.T, 𝒫.γ
Γ′ = [[R(s,a) + γ*sum(T(s,a,s′)*maximum(α′[j] for α′ in Γ)

for (j,s′) in enumerate(𝒮)) for s in 𝒮] for a in 𝒜]
return Γ′

end

function solve(M::QMDP, 𝒫::POMDP)
Γ = [zeros(length(𝒫.𝒮)) for a in 𝒫.𝒜]
Γ = alphavector_iteration(𝒫, M, Γ)
return AlphaVectorPolicy(𝒫, Γ, 𝒫.𝒜)

end

Algorithm 21.2. The QMDP al-
gorithm, which finds an approx-
imately optimal policy for an
infinite-horizon POMDP with a
discrete state and action space,
where k_max is the number of it-
erations. QMDP assumes perfect
observability.

Figure 21.1. Value functions ob-
tained for the crying baby problem
(appendix F.7) using QMDP. In the
first iteration, a single alpha vec-
tor dominates. In subsequent itera-
tions, two alpha vectors dominate.

When QMDP is run to the horizon in finite horizon problems or to convergence
for infinite-horizon problems, the resulting policy is equivalent to assuming that
there will be full observability after taking the first step. Because we can do better

21.2. fast informed bound 429

only if we have full observability, QMDP will produce an upper bound on the
true optimal value function U∗(b). In other words, maxa α

⊤
a b ≥ U∗(b) for all b.4 4 Although the value function rep-

resented by the QMDP alpha vec-
tors upper-bounds the optimal
value function, the utility realized
by a QMDP policy will not exceed
that of an optimal policy in expec-
tation, of course.

If QMDP is not run to convergence for infinite-horizon problems, it might
not provide an upper bound. One way to guarantee that QMDP will provide an
upper bound after a finite number of iterations is to initialize the value function
to some upper bound. One rather loose upper bound is the best-action best-state
upper bound, which is the utility obtained from taking the best action from the
best state forever:

U(b) = max
s,a

R(s, a)

1− γ
(21.2)

The assumption of full observability after the first step can cause QMDP to
poorly approximate the value of information-gathering actions, which are actions
that significantly reduce the uncertainty in the state. For example, looking over
one’s shoulder before changing lanes when driving is an information-gathering
action. QMDP can perform well in problems where the optimal policy does not
include costly information gathering.

We can generalize the QMDP approach to problems that may not have a small,
discrete state space. In such problems, the iteration in equation (21.1) may not be
feasible, but wemay use one of themanymethods discussed in earlier chapters for
obtaining an approximate action value function Q(s, a). This value function might
be defined over a high-dimensional, continuous state space using, for example, a
neural network representation. The value function evaluated at a belief point is,
then,

U(b) = max
a

∫

Q(s, a)b(s)ds (21.3)

The integral above may be approximated through sampling.

21.2 Fast Informed Bound

As with QMDP, the fast informed bound computes one alpha vector for each ac-
tion. However, the fast informed bound takes into account, to some extent, the
observation model.5 The iteration is

5 The relationship between QMDP
and the fast informed bound,
together with empirical results,
are discussed by M. Hauskrecht,
“Value-Function Approximations
for Partially Observable Markov
Decision Processes,” Journal of Ar-
tificial Intelligence Research, vol. 13,
pp. 33–94, 2000.

α
(k+1)
a (s) = R(s, a) + γ ∑

o

max
a′

∑
s′

O(o | a, s′)T(s′ | s, a)α
(k)
a′ (s

′) (21.4)

which requires O(|A|2|S|2|O|) operations per iteration.

430 chapter 21. offline belief state planning

The fast informed bound provides an upper bound on the optimal value
function. That upper bound is guaranteed to be no looser than that provided by
QMDP, and it also tends to be tighter. The fast informed bound is implemented
in algorithm 21.3 and is used in figure 21.2 to compute optimal value functions.

struct FastInformedBound
k_max # maximum number of iterations

end

function update(𝒫::POMDP, M::FastInformedBound, Γ)
𝒮, 𝒜, 𝒪, R, T, O, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒪, 𝒫.R, 𝒫.T, 𝒫.O, 𝒫.γ
Γ′ = [[R(s, a) + γ*sum(maximum(sum(O(a,s′,o)*T(s,a,s′)*α′[j]

for (j,s′) in enumerate(𝒮)) for α′ in Γ) for o in 𝒪)
for s in 𝒮] for a in 𝒜]

return Γ′
end

function solve(M::FastInformedBound, 𝒫::POMDP)
Γ = [zeros(length(𝒫.𝒮)) for a in 𝒫.𝒜]
Γ = alphavector_iteration(𝒫, M, Γ)
return AlphaVectorPolicy(𝒫, Γ, 𝒫.𝒜)

end

Algorithm 21.3. The fast informed
bound algorithm, which finds an
approximately optimal policy for
an infinite-horizon POMDP with
discrete state, action, and obser-
vation spaces, where k_max is the
number of iterations.

Figure 21.2. Value functions ob-
tained for the crying baby problem
using the fast informed bound. The
value function after 10 iterations is
noticeably lower than that of the
QMDP algorithm.

21.3 Fast Lower Bounds

The previous two sections introduced methods that can be used to produce upper
bounds on the value function represented as alpha vectors. This section introduces

21.4. point-based value iteration 431

a couple of methods for quickly producing lower bounds represented as alpha
vectors without any planning in the belief space. Although the upper-bound
methods can often be used directly to produce sensible policies, the lower bounds
discussed in this section are generally only used to seed other planning algorithms.
Figure 21.3 plots the two lower-bound methods discussed in this section.

Figure 21.3. Blind lower bounds
with different numbers of itera-
tions and the BAWS lower bound
applied to the crying baby prob-
lem.

A common lower bound is the best-action worst-state (BAWS) lower bound (algo-
rithm 21.4). It is the discounted reward obtained by taking the best action in the
worst state forever:

rbaws = max
a

∞

∑
k=1

γk−1 min
s

R(s, a) =
1

1− γ
max

a
min

s
R(s, a) (21.5)

This lower bound is represented by a single alpha vector. This bound is typically
very loose, but it can be used to seed other algorithms that can tighten the bound,
as we will discuss shortly.

function baws_lowerbound(𝒫::POMDP)
𝒮, 𝒜, R, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.R, 𝒫.γ
r = maximum(minimum(R(s, a) for s in 𝒮) for a in 𝒜) / (1-γ)
α = fill(r, length(𝒮))
return α

end

Algorithm 21.4. Implementation
of the best-action worst-state lower
bound from equation (21.5) repre-
sented as an alpha vector.

The blind lower bound (algorithm 21.5) represents a lower boundwith one alpha
vector per action. It makes the assumption that we are forced to commit to a single
action forever, blind to what we observe in the future. To compute these alpha
vectors, we start with another lower bound (typically the best-action worst-state
lower bound) and then perform a number of iterations:

α
(k+1)
a (s) = R(s, a) + γ ∑

s′
T(s′ | s, a)α

(k)
a (s′) (21.6)

This iteration is similar to the QMDP update in equation (21.1), except that it
does not have a maximization over the alpha vectors on the right-hand side.

21.4 Point-Based Value Iteration

QMDP and the fast informed bound generate one alpha vector for each action,
but the optimal value function is often better approximated by many more

432 chapter 21. offline belief state planning

function blind_lowerbound(𝒫, k_max)
𝒮, 𝒜, T, R, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.T, 𝒫.R, 𝒫.γ
Q(s,a,α) = R(s,a) + γ*sum(T(s,a,s′)*α[j] for (j,s′) in enumerate(𝒮))
Γ = [baws_lowerbound(𝒫) for a in 𝒜]
for k in 1:k_max

Γ = [[Q(s,a,α) for s in 𝒮] for (α,a) in zip(Γ, 𝒜)]
end
return Γ

end

Algorithm 21.5. Implementation of
the blind lower bound represented
as a set of alpha vectors.

alpha vectors. Point-based value iteration6 computes m different alpha vectors 6 A survey of point-based value it-
eration methods are provided by
G. Shani, J. Pineau, and R. Kaplow,
“A Survey of Point-Based POMDP
Solvers,” Autonomous Agents and
Multi-Agent Systems, vol. 27, pp. 1–
51, 2012. That reference provides a
slightly different way to compute a
belief backup, though the result is
the same.

Γ = {α1, . . . ,αm}, each associated with different belief points B = {b1, . . . , bm}.
Methods for selecting these beliefs will be discussed in section 21.7. As before,
these alpha vectors define an approximately optimal value function:

UΓ(b) = max
α∈Γ

α⊤b (21.7)

The algorithmmaintains a lower bound on the optimal value function,UΓ(b) ≤
U∗(b) for all b. We initialize our alpha vectors to start with a lower bound and
then perform a backup to update the alpha vectors at each point in B. The backup
operation (algorithm 21.6) takes a belief b and a set of alpha vectors Γ and
constructs a new alpha vector. The algorithm iterates through every possible
action a and observation o and extracts the alpha vector from Γ that is maximal at
the resulting belief state:

αa,o = arg max
α∈Γ

α⊤Update(b, a, o) (21.8)

Then, for each available action a, we construct a new alpha vector based on these
αa,o vectors:

αa(s) = R(s, a) + γ ∑
s′ ,o

O(o | a, s′)T(s′ | s, a)αa,o(s
′) (21.9)

The alpha vector that is ultimately produced by the backup operation is

α = arg max
αa

α⊤a b (21.10)

If Γ is a lower bound, the backup operation will produce only alpha vectors that
are also a lower bound.

21.5. randomized point-based value iteration 433

Repeated application of the backup operation over the beliefs in B gradually
increases the lower bound on the value function represented by the alpha vectors
until convergence. The converged value function will not necessarily be optimal
because B typically does not include all beliefs reachable from the initial belief.
However, so long as the beliefs in B are well distributed across the reachable belief
space, the approximation may be acceptable. In any case, the resulting value
function is guaranteed to provide a lower bound that can be used with other
algorithms, potentially online, to further improve the policy.

Point-based value iteration is implemented in algorithm 21.7. Figure 21.4 shows
several iterations on an example problem.

function backup(𝒫::POMDP, Γ, b)
𝒮, 𝒜, 𝒪, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒪, 𝒫.γ
R, T, O = 𝒫.R, 𝒫.T, 𝒫.O
Γa = []
for a in 𝒜

Γao = []
for o in 𝒪

b′ = update(b, 𝒫, a, o)
push!(Γao, argmax(α->α⋅b′, Γ))

end
α = [R(s, a) + γ*sum(sum(T(s, a, s′)*O(a, s′, o)*Γao[i][j]

for (j,s′) in enumerate(𝒮)) for (i,o) in enumerate(𝒪))
for s in 𝒮]

push!(Γa, α)
end
return argmax(α->α⋅b, Γa)

end

Algorithm21.6. Amethod for back-
ing up a belief for a POMDP with
discrete state and action spaces,
where Γ is a vector of alpha vectors
and b is a belief vector at which
to apply the backup. The update
method for vector beliefs is defined
in algorithm 19.2.

21.5 Randomized Point-Based Value Iteration

Randomized point-based value iteration (algorithm 21.8) is a variation of the point-
based value iteration approach from the previous section.7 The primary difference 7 M.T. J. Spaan and N.A. Vlassis,

“Perseus: Randomized Point-Based
Value Iteration for POMDPs,” Jour-
nal of Artificial Intelligence Research,
vol. 24, pp. 195–220, 2005.

is the fact that we are not forced to maintain an alpha vector at every belief in B.
We initialize the algorithm with a single alpha vector in Γ, and then update Γ at
every iteration, potentially increasing or decreasing the number of alpha vectors
in Γ as appropriate. This modification of the update step can improve efficiency.

434 chapter 21. offline belief state planning

struct PointBasedValueIteration
B # set of belief points
k_max # maximum number of iterations

end

function update(𝒫::POMDP, M::PointBasedValueIteration, Γ)
return [backup(𝒫, Γ, b) for b in M.B]

end

function solve(M::PointBasedValueIteration, 𝒫)
Γ = fill(baws_lowerbound(𝒫), length(𝒫.𝒜))
Γ = alphavector_iteration(𝒫, M, Γ)
return LookaheadAlphaVectorPolicy(𝒫, Γ)

end

Algorithm 21.7. Point-based value
iteration, which finds an approx-
imately optimal policy for an
infinite-horizon POMDP with dis-
crete state, action, and observation
spaces, where B is a vector of be-
liefs and k_max is the number of it-
erations.

Figure 21.4. Approximate value
functions obtained using point-
based value iteration on the cry-
ing baby problem with belief vec-
tors [1/4, 3/4] and [3/4, 1/4]. Un-
like QMDP and the fast informed
bound, the value function of point-
based value iteration is always a
lower bound of the true value func-
tion.

21.5. randomized point-based value iteration 435

Each update takes a set of alpha vectors Γ as input and outputs a set of alpha
vectors Γ′ that improve on the value function represented by Γ at the beliefs in B.
In other words, it outputs Γ′ such that UΓ′(b) ≥ UΓ(b) for all b ∈ B. We begin
by initializing Γ′ to the empty set and initializing B′ to B. We then remove a point
b randomly from B′ and perform a belief backup (algorithm 21.6) on b, using
Γ to get a new alpha vector, α. We then find the alpha vector in Γ ∪ {α} that
dominates at b and add it to Γ′. All belief points in B′ whose value is improved
with this alpha vector is then removed from B′. As the algorithm progresses, B′

becomes smaller and contains the set of points that have not been improved by
Γ′. The update finishes when B′ is empty. Figure 21.5 illustrates this process with
the crying baby problem.

struct RandomizedPointBasedValueIteration
B # set of belief points
k_max # maximum number of iterations

end

function update(𝒫::POMDP, M::RandomizedPointBasedValueIteration, Γ)
Γ′, B′ = [], copy(M.B)
while !isempty(B′)

b = rand(B′)
α = argmax(α->α⋅b, Γ)
α′ = backup(𝒫, Γ, b)
if α′⋅b ≥ α⋅b

push!(Γ′, α′)
else

push!(Γ′, α)
end
filter!(b->maximum(α⋅b for α in Γ′) <

maximum(α⋅b for α in Γ), B′)
end
return Γ′

end

function solve(M::RandomizedPointBasedValueIteration, 𝒫)
Γ = [baws_lowerbound(𝒫)]
Γ = alphavector_iteration(𝒫, M, Γ)
return LookaheadAlphaVectorPolicy(𝒫, Γ)

end

Algorithm 21.8. Randomized
point-based value iteration, which
finds an approximately optimal
policy for an infinite-horizon
POMDP with discrete state, action,
and observation spaces, where B is
a vector of beliefs and k_max is the
number of iterations.

Figure 21.5. Approximate value
functions obtained using random-
ized point-based value iteration
on the crying baby problem with
belief points at [1/4, 3/4] and
[3/4, 1/4].

21.6 Sawtooth Upper Bound

The sawtooth upper bound is an alternative way to represent the value function.
Instead of storing a set of alpha vectors Γ, we store a set of belief-utility pairs:

V = {(b1, u1), . . . , (bm, um)} (21.11)

where ui = U(bi). These pairs must include all the standard basis beliefs:

E = {e1 = [1, 0, . . . , 0], . . . , en = [0, 0, . . . , 1]} (21.12)

In other words,
{(e1, U(e1)), . . . , (en, U(en))} ⊆ V (21.13)

If these utilities are upper bounds (e.g., as obtained from the fast informed bound),
then the way that we use V to estimate U(b) at an arbitrary belief b will result in
an upper bound.8 8 The relationship between saw-

tooth and other bounds are dis-
cussed by M. Hauskrecht, “Value-
Function Approximations for Par-
tially Observable Markov Decision
Processes,” Journal of Artificial Intel-
ligence Research, vol. 13, pp. 33–94,
2000.

The ‘‘sawtooth’’ name comes from the way that we estimate U(b) by interpolat-
ing points inV. For each belief-utility pair (b, U(b)) inV, we forma single, pointed
‘‘tooth.’’ When multiple pairs are considered, it forms a ‘‘sawtooth’’ shape. If the
belief space is n-dimensional, each tooth is an inverted, n-dimensional pyramid.
The base of each pyramid is formed by the standard basis beliefs (ei, U(ei)). The
apex of each pyramid corresponds to one of the belief-utility pairs (b, U(b)) ∈ V.
The walls of each pyramid can be defined by hyperplanes. The combination of
multiple pyramids forms the n-dimensional sawtooth. The sawtooth upper bound
at any belief is the minimum value among these pyramids at that belief.

21.6. sawtooth upper bound 437

Consider the sawtooth representation in a two-state POMDP, such as in the
crying baby problem as shown in figure 21.6. The corners of each tooth are the
values U(e1) and U(e2) for each standard basis belief ei. The sharp lower point
of each tooth is the value U(b), since each tooth is a point-set pair (b, U(b)). The
linear interpolation from U(e1) to U(b), and again from U(b) to U(e2), form
the tooth. To combine multiple teeth and form the upper bound, we take the
minimum interpolated value at any belief, creating the distinctive sawtooth shape.
As is apparent in figure 21.6, the sawtooth upper bound is a piecewise linear
function, but it is not convex, in contrast with the alpha vector representations
discussed earlier in this chapter.

0 0.2 0.4 0.6 0.8 1
−15

−10

−5

0

P(hungry)

U
(b
)

Figure 21.6. The sawtooth upper
bound representation applied to
the crying baby problem.

We can entend the intuition from the two-state case to allow us to compute the
sawtooth upperbound UV(b) for any belief b over an arbitrary number of states
from the set of pairs in V. First, wewill define how to compute the upperbound for
a single tooth associated with a nonbasis pair (b′, u′) in V. In order to define this
upperbound U(b′ ,u′)(b), we will use uE = [U(e1), . . . , U(en)], which is a vector
containing the utilities at all of the basis beliefs. We will also use what is called
the minimum ratio:

ρ(b, b′) = min
i

bi

b′i
(21.14)

It turns out that the upperbound can be written

U(b′ ,u′)(b) = u⊤E b + ρ(b, b′)(u′ − u⊤E b′) (21.15)

To compute the sawtooth upperbound with all the teeth defined by V, we take
the minimum over all the teeth:

UV(b) = min
(b′ ,u′)∈V|b′ 6∈E

U(b′ ,u′)(b) (21.16)

Algorithm 21.9 provides an implementation. We can also derive a policy using
greedy one-step lookahead.

We can iteratively apply greedy one-step lookahead at a set of beliefs B to
tighten our estimates of the upper bound. The beliefs in B can be a superset of the
beliefs in V. Algorithm 21.10 provides an implementation of this. Example 21.1
shows the effect of multiple iterations of the sawtooth approximation on the
crying baby problem.

struct SawtoothPolicy
𝒫 # POMDP problem
V # dictionary mapping beliefs to utilities

end

function basis(𝒫)
n = length(𝒫.𝒮)
e(i) = [j == i ? 1.0 : 0.0 for j in 1:n]
return [e(i) for i in 1:n]

end

function utility(π::SawtoothPolicy, b)
𝒫, V = π.𝒫, π.V
if haskey(V, b)

return V[b]
end
uE = [V[e] for e in E]
ρ(b′) = minimum(b[i] / b′[i] for i in eachindex(b) if b′[i] > 0)
U(b′, u′) = uE⋅b + ρ(b′)*(u′ - uE⋅b′)
return minimize(U(b′, u′) for (b′, u′) in V if b′ ∉ E)

end

(π::SawtoothPolicy)(b) = greedy(π, b).a

Algorithm 21.9. The sawtooth up-
per bound representation for value
functions and policies. It is defined
using a dictionary V that maps be-
lief vectors to upper bounds on
their utility obtained, such as, from
the fast informed bound. A require-
ment of this representation is that
V contain belief-utility pairs at the
standard basis beliefs, which can
be obtained from the basis func-
tion. We can use one-step looka-
head to obtain greedy action-utility
pairs from arbitrary beliefs b.

struct SawtoothIteration
V # initial mapping from beliefs to utilities
B # beliefs to compute values including those in V map
k_max # maximum number of iterations

end

function solve(M::SawtoothIteration, 𝒫::POMDP)
E = basis(𝒫)
π = SawtoothPolicy(𝒫, M.V)
for k in 1:M.k_max

V = Dict(b => (b ∈ E ? M.V[b] : greedy(π, b).u) for b in M.B)
π = SawtoothPolicy(𝒫, V)

end
return π

end

Algorithm 21.10. Sawtooth iter-
ation iteratively applies one-step
lookahead at points in B to improve
the utility estimates at the points in
V. The beliefs in B are a superset of
those contained in V. To preserve
the upper bound at each iteration,
updates are not made at the stan-
dard basis beliefs stored in E. We
run k_max iterations.

Suppose that we want to maintain an upper bound of the value for the crying
baby problem with regularly spaced belief points with a step size of 0.2. To
obtain an initial upper bound, we use the fast informed bound. We can then
run sawtooth iteration for three steps as follows:
n = length(𝒫.𝒮)
πfib = solve(FastInformedBound(1), 𝒫)
V = Dict(e => utility(πfib, e) for e in basis(𝒫))
B = [[p, 1 - p] for p in 0.0:0.2:1.0]
π = solve(SawtoothIteration(V, B, 2), 𝒫)

The sawtooth upper bound improves as follows:

Example 21.1. An illustration of
sawtooth’s ability to maintain an
upper bound at regularly spaced
beliefs for the crying baby prob-
lem.

21.7 Point Selection

Algorithms like point-based value iteration and sawtooth iteration require a set
of beliefs B. We want to choose B so that there are more points in the relevant
areas of the belief space; we do not want to waste computation on beliefs that
we are not likely to reach under our (hopefully approximately optimal) policy.
One way to explore the potentially reachable space is to take steps in the belief
space (algorithm 21.11). The outcome of the step will be random because the
observation is generated according to our probability model.

function randstep(𝒫::POMDP, b, a)
s = rand(SetCategorical(𝒫.𝒮, b))
s′, r, o = 𝒫.TRO(s, a)
b′ = update(b, 𝒫, a, o)
return b′, r

end

Algorithm 21.11. A function for
randomly sampling the next belief
b′ and reward r, given the current
belief b and action a in problem 𝒫.

We can create B from the belief states reachable from some initial belief under
a random policy. This random belief expansion procedure (algorithm 21.12) may
explore much more of the belief space than might be necessary; the belief space
reachable by a random policy can be much larger than the space reachable by
an optimal policy. Of course, computing the belief space that is reachable by an
optimal policy generally requires knowing the optimal policy, which is what
we want to compute in the first place. One approach that can be taken is to use
successive approximations of the optimal policy to iteratively generate B.9 9 This is the intuition behind the

algorithm known as Successive Ap-
proximations of the Reachable Space
under Optimal Policies (SARSOP).
H. Kurniawati, D. Hsu, and W. S.
Lee, “SARSOP: Efficient Point-
Based POMDP Planning by Ap-
proximating Optimally Reachable
Belief Spaces,” in Robotics: Science
and Systems, 2008.

In addition to wanting our belief points to be focused on the reachable belief
space, we want those points to be spread out to allow better value function
approximation. The quality of the approximation provided by the alpha vectors
associated with the points in B degrades as we evaluate points farther from B.
We can take an exploratory belief expansion approach (algorithm 21.13), where
we try every action for every belief in B and add the resulting belief states that
are farthest from the beliefs already in the set. Distance in belief space can be
measured in different ways. This algorithm uses the L1-norm.10 Figure 21.7 shows

10 The L1 distance between b and b′

is ∑s |b(s)− b′(s)| and is denoted
as ‖b− b′‖1. See appendix A.4.an example of the belief points added to B using this approach.

function random_belief_expansion(𝒫, B)
B′ = copy(B)
for b in B

a = rand(𝒫.𝒜)
b′, r = randstep(𝒫, b, a)
push!(B′, b′)

end
return unique!(B′)

end

Algorithm 21.12. Randomly ex-
panding a finite set of beliefs B
used in point-based value iteration
based on reachable beliefs.

function exploratory_belief_expansion(𝒫, B)
B′ = copy(B)
for b in B

best = (b=copy(b), d=0.0)
for a in 𝒫.𝒜

b′, r = randstep(𝒫, b, a)
d = minimum(norm(b - b′, 1) for b in B′)
if d > best.d

best = (b=b′, d=d)
end

end
push!(B′, best.b)

end
return unique!(B′)

end

Algorithm 21.13. Expanding a fi-
nite set of beliefs B used in point-
based value iteration by exploring
the reachable beliefs and adding
those that are farthest from the cur-
rent beliefs.

Figure 21.7. Exploratory belief ex-
pansion run on the three-state ma-
chine replacement problem, start-
ing with an initial uniform be-
lief b = [1/3, 1/3, 1/3]. New be-
liefs were added if the distance to
any previous belief was at least
0.05.

21.8 Sawtooth Heuristic Search

Chapter 9 introduced the concept of heuristic search as an online method in the
fully observable context. This section discusses sawtooth heuristic search (algo-
rithm 21.14) as an offline method that produces a set of alpha vectors that can be
used to represent an offline policy. However, like the online POMDP methods
discussed in the next chapter, the computational effort is focused on beliefs that
are reachable from some specified initial belief. The heuristic that drives the ex-
ploration of the reachable belief space is the gap between the upper and lower
bounds of the value function.11

11 The heuristic search value iteration
(HSVI) algorithm introduced the
concept of using the sawtooth-
based action heuristic and
gap-based observation heuristic.
T. Smith and R.G. Simmons,
“Heuristic Search Value Iteration
for POMDPs,” in Conference on
Uncertainty in Artificial Intelli-
gence (UAI), 2004. The SARSOP
algorithm built on this work. H.
Kurniawati, D. Hsu, and W. S. Lee,
“SARSOP: Efficient Point-Based
POMDP Planning by Approximat-
ing Optimally Reachable Belief
Spaces,” in Robotics: Science and
Systems, 2008.

The algorithm is initialized with an upper bound on the value function rep-
resented by a set of sawtooth belief-utility pairs V, together with a lower bound
on the value function represented by a set of alpha vectors Γ. The belief-utility
pairs defining the sawtooth upper bound can be obtained from the fast informed
bound. The lower bound can be obtained from the best-action worst-state bound,
as shown in algorithm 21.14, or some other method, such as point-based value
iteration.

struct SawtoothHeuristicSearch
b # initial belief
δ # gap threshold
d # depth
k_max # maximum number of iterations
k_fib # number of iterations for fast informed bound

end

function explore!(M::SawtoothHeuristicSearch, 𝒫, πhi, πlo, b, d=0)
𝒮, 𝒜, 𝒪, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒪, 𝒫.γ
ϵ(b′) = utility(πhi, b′) - utility(πlo, b′)
if d ≥ M.d || ϵ(b) ≤ M.δ / γ^d

return
end
a = πhi(b)
o = argmax(o -> ϵ(update(b, 𝒫, a, o)), 𝒪)
b′ = update(b, 𝒫, a, o)
explore!(M, 𝒫, πhi, πlo, b′, d+1)
if b′ ∉ basis(𝒫)

πhi.V[b′] = greedy(πhi, b′).u
end
push!(πlo.Γ, backup(𝒫, πlo.Γ, b′))

end

function solve(M::SawtoothHeuristicSearch, 𝒫::POMDP)
πfib = solve(FastInformedBound(M.k_fib), 𝒫)
Vhi = Dict(e => utility(πfib, e) for e in basis(𝒫))
πhi = SawtoothPolicy(𝒫, Vhi)
πlo = LookaheadAlphaVectorPolicy(𝒫, [baws_lowerbound(𝒫)])
for i in 1:M.k_max

explore!(M, 𝒫, πhi, πlo, M.b)
if utility(πhi, M.b) - utility(πlo, M.b) < M.δ

break
end

end
return πlo

end

Algorithm 21.14. The sawtooth
heuristic search policy. The solver
starts from belief b and explores to
a depth d for no more than k_max
iterations. It uses an upper bound
obtained through the fast informed
bound computed with k_fib it-
erations. The lower bound is ob-
tained from the best-action worst-
state bound. The gap threshold is
δ.

At each iteration, we explore beliefs that are reachable from our initial belief
to a maximum depth. As we explore, we update the set of belief-action pairs
forming our sawtooth upper bound and the set of alpha vectors forming our
lower bound. We stop exploring after a certain number of iterations or until the
gap at our initial state is below a threshold δ > 0.

When we encounter a belief b along our path from the initial node during our
exploration, we check whether the gap at b is below a threshold δ/γd, where d

is our current depth. If we are below that threshold, then we can stop exploring
along that branch. We want the threshold to increase as d increases because the
gap at b after an update is at most γ times the weighted average of the gap at the
beliefs that are immediately reachable.

If the gap at b is above the threshold and we have not reached our maximum
depth, then we can explore the next belief, b′. First, we determine the action a

recommended by our sawtooth policy. Then, we choose the observation o that
maximizes the gap at the resulting belief.12 We recursively explore down the tree. 12 Some variants simply sample the

next observations. Others select
the observation that maximizes the
gap weighted by its likelihood.

After exploring the descendants of b′, we add (b′, u) to V, where u is the one-step
lookahead value of b′. We add to Γ the alpha vector that results from a backup at
b′. Figure 21.8 shows the tightening of the bounds.

Figure 21.8. The evolution of the
upper bound, represented by saw-
tooth pairs, and the lower bound,
represented by alpha vectors for
the crying baby problem. The op-
timal value function is shown in
black.

21.9 Triangulated Value Functions

As discussed in section 20.1, a POMDP can be converted to a belief-state MDP.
The state space in that belief-state MDP is continuous, corresponding to the
space of possible beliefs in the original POMDP. We can approximate the value
function in a way similar to what was described in chapter 8 and then apply a
dynamic programming algorithm such as value iteration to the approximation.
This section discusses a particular kind of local value function approximation
that involves Freudenthal triangulation13 over a discrete set of belief points B. This

13 H. Freudenthal, “Simplizialzer-
legungen von Beschränkter Flach-
heit,”Annals of Mathematics, vol. 43,
pp. 580–582, 1942. This triangu-
lation method was applied to
POMDPs in W. S. Lovejoy, “Com-
putationally Feasible Bounds for
Partially Observed Markov De-
cision Processes,” Operations Re-
search, vol. 39, no. 1, pp. 162–175,
1991.triangulation allows us to interpolate the value function at arbitrary points in the

belief space. As with the sawtooth representation, we use a set of belief-utility
pairs V = {(b, U(b)) | b ∈ B} to represent our value function. This approach can
be used to obtain an upper bound on the value function.

Freudenthal interpolation in belief space involves spreading the belief points
in B evenly over the space, as shown in figure 21.9. The number of beliefs in B

depends on the dimensionality n and granularity m of the Freudenthal triangula-
tion:14

14 FreudenthalTriangulations.jl
provides an implementation for
generating these beliefs.

|B| = (m + n− 1)!

m!(n− 1)!
(21.17)

0

0.2

0.4

0.6

0.8

1

0 0.2 0.4 0.6 0.8 1
0

0.2

0.4

0.6

0.8

1

P
(s 1

)P
(s

2 )

P(s3)

Figure 21.9. A belief state dis-
cretization using Freudenthal trian-
gulation in n = 3-dimensional be-
lief space with granularity m = 10.

We can estimate U(b) at an arbitrary point b by interpolating values at the
discrete points in B. Similar to the simplex interpolation introduced in section 8.5,
we find the set of belief points in B that form a simplex that encloses b and weight
their values together. In n-dimensional belief spaces, there are up to n + 1 vertices
whose values need to be weighted together. If b(1), . . . , b(n+1) are the enclosing
points and λ1, . . . , λn+1 are their weights, then the estimate of the value at b is

U(b) = ∑
i

λiU(b(i)) (21.18)

Algorithm 21.15 extracts this utility function and policy from the pairs in V.
Algorithm 21.16 applies a variation of approximate value iteration (introduced

in algorithm 8.1) to our triangulated policy representation. We simply iteratively
apply backups over our beliefs in B using one-step lookahead with our value
function interpolation. If U is initialized with an upper bound, value iteration will
result in an upper bound even after a finite number of iterations. This property
holds because value functions are convex and the linear interpolation between ver-
tices on the value function must lie on or above the underlying convex function.15

15 See lemma 4 of W. S. Lovejoy,
“Computationally Feasible Bounds
for Partially Observed Markov De-
cision Processes,” Operations Re-
search, vol. 39, no. 1, pp. 162–175,
1991.Figure 21.10 shows an example of a policy and utility function.

struct TriangulatedPolicy
𝒫 # POMDP problem
V # dictionary mapping beliefs to utilities
B # beliefs
T # Freudenthal triangulation

end

function TriangulatedPolicy(𝒫::POMDP, m)
T = FreudenthalTriangulation(length(𝒫.𝒮), m)
B = belief_vertices(T)
V = Dict(b => 0.0 for b in B)
return TriangulatedPolicy(𝒫, V, B, T)

end

function utility(π::TriangulatedPolicy, b)
B, λ = belief_simplex(π.T, b)
return sum(λi*π.V[b] for (λi, b) in zip(λ, B))

end

(π::TriangulatedPolicy)(b) = greedy(π, b).a

Algorithm 21.15. A policy rep-
resentation using Freudenthal
triangulation with granularity m.
As with the sawtooth method,
we maintain a dictionary that
maps belief vectors to utilities.
This implementation initializes
the utilities to 0, but if we want to
represent an upper bound, then
we would need to initialize those
utilities appropriately. We define a
function to estimate the utility of
a given belief using interpolation.
We can extract a policy using
greedy lookahead. The Freuden-
thal triangulation structure is
passed the dimensionality and
granularity at construction. The
FreudenthalTriangulations.jl
package provides the function
belief_vertices, which returns
B, given a particular triangulation.
It also provides belief_simplex,
which returns the set of enclosing
points and weights for a belief.

struct TriangulatedIteration
m # granularity
k_max # maximum number of iterations

end

function solve(M::TriangulatedIteration, 𝒫)
π = TriangulatedPolicy(𝒫, M.m)
U(b) = utility(π, b)
for k in 1:M.k_max

U′ = [greedy(𝒫, U, b).u for b in π.B]
for (b, u′) in zip(π.B, U′)

π.V[b] = u′
end

end
return π

end

Algorithm 21.16. Approximate
value iteration with k_max itera-
tions using a triangulated policy
with granularity m. At each itera-
tion, we update the utilities asso-
ciated with the beliefs in B using
greedy one-step lookahead with
triangulated utilities.

Figure 21.10. The policy and value
function for the maintenance prob-
lem with granularity m = 10 af-
ter 11 iterations. The value func-
tion plot shows the discrete belief
points aswhite dots. This policy ap-
proximates the exact policy given
in appendix F.8.

21.10 Summary

• TheQMDPalgorithmassumes perfect observability after the first step, resulting
in an upper bound on the true value function.

• The fast informed bound provides a tighter upper bound on the value function
than QMDP by accounting for the observation model.

• Point-based value iteration provides a lower bound on the value function using
alpha vectors at a finite set of beliefs.

• Randomized point-based value iteration performs updates at randomly se-
lected points in the belief set until the values at all points in the set are improved.

• The sawtooth upper bound allows iterative improvement of the fast informed
bound using an efficient point-set representation.

• Carefully selecting which belief points to use in point-based value iteration
can improve the quality of the resulting policies.

• Sawtooth heuristic search attempts to tighten the upper and lower bounds of the
value function represented by sawtooth pairs and alpha vectors, respectively.

• One approach to approximately solving POMDPs is to discretize the belief
space, and then to apply dynamic programming to extract an upper bound on
the value function and a policy.

21.11 Exercises
Exercise 21.1. Suppose that we are in a variation of the straight-line hex world problem
(appendix F.1) consisting of four cells corresponding to states s1:4. There are two actions:
move left (ℓ) and move right (r). The effects of those actions are deterministic. Moving
left in s1 or moving right in s4 gives a reward of 100 and ends the game. With a discount
factor of 0.9, compute alpha vectors using QMDP. Then, using the alpha vectors, compute
the approximately optimal action, given the belief b = [0.3, 0.1, 0.5, 0.1].

Solution:We denote the alpha vector associated with moving left as αℓ and the alpha vector
associated with moving right as αr. We initialize the alpha vectors to zero:

α
(1)
ℓ

= [R(s1, ℓ), R(s2, ℓ), R(s3, ℓ), R(s4, ℓ)] = [0, 0, 0, 0]

α
(1)
r = [R(s1, r), R(s2, r), R(s3, r), R(s4, r)] = [0, 0, 0, 0]

In the first iteration, since all the entries in the alpha vectors are zero, only the reward term
contributes to the QMDP update (equation (21.1)):

α
(2)
ℓ

= [100, 0, 0, 0]

α
(2)
r = [0, 0, 0, 100]

In the next iteration, we apply the update, which leads to new values for s2 for the left
alpha vector and for s3 for the right alpha vector. The updates for the left alpha vector are
as follows (with the right alpha vector updates being symmetric):

α
(3)
ℓ

(s1) = 100 (terminal state)
α
(3)
ℓ

(s2) = 0 + 0.9×max(α
(2)
ℓ

(s1), α
(2)
r (s1)) = 90

α
(3)
ℓ

(s3) = 0 + 0.9×max(α
(2)
ℓ

(s2), α
(2)
r (s2)) = 0

α
(3)
ℓ

(s4) = 0 + 0.9×max(α
(2)
ℓ

(s3), α
(2)
r (s3)) = 0

This leads to the following:

α
(3)
ℓ

= [100, 90, 0, 0]

α
(3)
r = [0, 0, 90, 100]

© 2022 Massachusetts Institute of Technology, shared under a Creative Commons CC-BY-NC-ND license.
2024-02-06 20:54:49-08:00, comments to bugs@algorithmsbook.com



21.11. exercises 449

In the third iteration, the updates for the left alpha vector are

α
(4)
ℓ

(s1) = 100 (terminal state)
α
(4)
ℓ

(s2) = 0 + 0.9×max(α
(3)
ℓ

(s1), α
(3)
r (s1)) = 90

α
(4)
ℓ

(s3) = 0 + 0.9×max(α
(3)
ℓ

(s2), α
(3)
r (s2)) = 81

α
(4)
ℓ

(s4) = 0 + 0.9×max(α
(3)
ℓ

(s3), α
(3)
r (s3)) = 81

Our alpha vectors are, then,

α
(4)
ℓ

= [100, 90, 81, 81]

α
(4)
r = [81, 81, 90, 100]

At this point, our alpha vector estimates have converged. We now determine the optimal
action by maximizing the utility associated with our belief over all actions:

α⊤
ℓ

b = 100× 0.3 + 90× 0.1 + 81× 0.5 + 81× 0.1 = 87.6

α⊤r b = 81× 0.3 + 81× 0.1 + 90× 0.5 + 100× 0.1 = 87.4

Thus, we find that moving left is the optimal action for this belief state, despite a higher
probability of being on the right half of the grid world. This is due to the relatively high
likelihood that we assign to being in state s1, where we would receive a large, immediate
reward by moving left.

Exercise 21.2. Recall the simplified hex world problem from exercise 21.1. Compute alpha
vectors for each action using the blind lower bound. Then, using the alpha vectors, compute
the value at the belief b = [0.3, 0.1, 0.5, 0.1].

Solution: The blind lower bound, shown in equation (21.6), is like the QMDP update, but
it lacks the maximization. We initialize the components of the alpha vectors to zero and
run to convergence as follows:

α
(2)
ℓ

= [100, 0, 0, 0]

α
(2)
r = [0, 0, 0, 100]

α
(3)
ℓ

= [100, 90, 0, 0]

α
(3)
r = [0, 0, 90, 100]

α
(4)
ℓ

= [100, 90, 81, 0]

α
(4)
r = [0, 81, 90, 100]

α
(5)
ℓ

= [100, 90, 81, 72.9]

α
(5)
r = [72.9, 81, 90, 100]

At this point, our alpha vector estimates have converged. We now determine the value by
maximizing the utility associated with our belief over all actions:

α⊤
ℓ

b = 100× 0.3 + 90× 0.1 + 81× 0.5 + 72.9× 0.1 = 86.79

α⊤r b = 72.9× 0.3 + 81× 0.1 + 90× 0.5 + 100× 0.1 = 84.97

Thus, the lower bound at b is 86.79.

Exercise 21.3. What is the complexity of a backup at a single belief point in point-based
value iteration assuming that |Γ| > |S|?

Solution: In the process of doing a backup, we compute an αa,o for every action a and
observation o. Computing αa,o in equation (21.8) requires finding the alpha vector α
in Γ that maximizes α⊤Update(b, a, o). A belief update, as shown in equation (19.7), is
O(|S|2) because it iterates over all initial and successor states. Hence, computing αa,o

requires O(|Γ||S|+ |S|2) = O(|Γ||S|) operations for a specific a and o, resulting in a total
of O(|Γ||S||A||O|) operations. We then compute αa in equation (21.9) for every action a

using these values forαa,o, requiring in a total of O(|S|2|A||O|). Finding the alpha vectorαa

that maximizes α⊤a b requires O(|S||A|) operations once we have the αa values. Together,
we have O(|Γ||S||A||O|) operations for a backup at belief b.

Exercise 21.4. Consider the set of belief-utility pairs given by

V = {([1, 0], 0), ([0, 1],−10), ([0.8, 0.2],−4), ([0.4, 0.6],−6)}

Using weights wi = 0.5 for all i, determine the utility for belief b = [0.5, 0.5] using the
sawtooth upper bound.

Solution: We interpolate with the belief-utility pairs. For each nonbasis belief, we start by
finding the farthest basis belief, ei. Starting with b3, we compute as follows:

i3 = arg max
j

∥

∥

∥b− ej

∥

∥

∥

1
−
∥

∥

∥b3 − ej

∥

∥

∥

1

‖b− e1‖1 − ‖b3 − e1‖1 =

∥

∥

∥

∥

∥

[

0.5

0.5

]

−
[

1

0

]∥

∥

∥

∥

∥

1

−
∥

∥

∥

∥

∥

[

0.8

0.2

]

−
[

1

0

]∥

∥

∥

∥

∥

1

=

∥

∥

∥

∥

∥

[

−0.5

0.5

]∥

∥

∥

∥

∥

1

−
∥

∥

∥

∥

∥

[

−0.2

0.2

]∥

∥

∥

∥

∥

1

= 0.6

‖b− e2‖1 − ‖b3 − e2‖1 =

∥

∥

∥

∥

∥

[

0.5

0.5

]

−
[

0

1

]∥

∥

∥

∥

∥

1

−
∥

∥

∥

∥

∥

[

0.8

0.2

]

−
[

0

1

]∥

∥

∥

∥

∥

1

=

∥

∥

∥

∥

∥

[

0.5

−0.5

]∥

∥

∥

∥

∥

1

−
∥

∥

∥

∥

∥

[

0.8

−0.8

]∥

∥

∥

∥

∥

1

= −0.6

i3 = 1

Thus, e1 is the farthest basis belief from b3.
For b4, we compute the following:

i4 = arg max
j

∥

∥

∥b− ej

∥

∥

∥

1
−
∥

∥

∥b4 − ej

∥

∥

∥

1

∥

∥

∥

∥

∥

∥

0

∥

∥

∥

1

∥

∥

∥

∥

1

= 0.2

i4 = 2

Thus, e2 is the farthest basis belief from b4.

We can compute U(b) using our weights, along with the appropriate corners and utility
pairs (e2, b3) and (e1, b4):

U3(b) = 0.5×−4 + 0.5× (−10) = −7

U4(b) = 0.5×−6 + 0.5× 0 = −3

Finally, we compute U(b) by taking the minimum of U3(b) and U4(b). Thus, U(b) = −7.

Exercise 21.5. Suppose that we have a valid lower bound represented as a set of alpha
vectors Γ. Is it possible for a backup at a belief state b to result in an alpha vector α′, such
that α′⊤b is lower than the utility function represented by Γ? In other words, can a backup
at a belief b result in an alpha vector that assigns a lower utility to b than the value function
represented by Γ?

0 0.5 1
−2

−1

0

1

2

P(s1)

U
(b
)

Figure 21.11. An example of how
a backup at a belief can result in
an alpha vector that, on its own,
lowers the value at that belief com-
pared to the original value func-
tion. The belief b where we do the
update corresponds to P(s1) = 0.5.
The original value function, repre-
sented by Γ, is shown in red. The al-
pha vector resulting from a backup
at b is shown in blue.

Solution: It is possible. Suppose we have only one action, observations are perfect, there is
no discounting, and the state space is {s0, s1}. The reward is R(si) = i for all i, and states
transition deterministically to s0. We start with a valid lower bound, Γ = {[−1,+1]}, as
shown in red in figure 21.11. We choose b = [0.5, 0.5] for the belief where we do the backup.
Using equation (21.9), we obtain

α(s0) = R(s0) + UΓ(s0) = 0 + (−1) = −1

α(s1) = R(s1) + UΓ(s0) = 1 + (−1) = 0

Hence, the alpha vector that we get after a backup is [−1, 0], shown in blue in figure 21.11.
The utility at b with that alpha vector is −0.5. However, UΓ(b) = 0, showing that backing
up a belief can result in an alpha vector that represents a lower utility at that belief.
This fact motivates the use of the if statement in randomized point-based value iteration
(algorithm 21.8). That if statement will use either the alpha vector from the backup or the
dominating alpha vector in Γ at belief b, whichever gives the greatest utility estimate.

